#1/ CONCATENATING / REARRANGING / INTERLEAVING

The following data steps create three data sets and each data set has an ID and cholesterol ... 

```sas
data jan;
  input id : $3. chol @@;
datalines;
  123 240 234 210 345 200 456 250;
run;

data feb;
  input id : $3. chol @@;
datalines;
  234 200 345 190 456 220 567 300;
run;

data mar;
  input id : $3. chol @@;
datalines;
  123 230 234 185 345 190 456 200 567 290;
run;
```

A/ Now that you have the three data sets (JAN, FEB, MAR), write a SAS job that will combine all three data sets into one data set that has 13 observations and 3: ID, CHOL, MONTH (month should have the value 1 for January, 2 for February, 3 for March).

```sas
data jan_mar;
  set jan (in=_j) feb (in=_f) mar;
  if _j then month = '1';
  else if _f then month = '2';
  else month = '3';
run;
```

B/ Using the data set created in part A, make a new data set with five observations (one for each ID) and four variables: ID, CHOL1, CHOL2, CHOL3.

```sas
proc sort data=jan_mar;
  by id;
run;

proc transpose data=jan_mar out=cholesterol (drop=_:) prefix=chol;
  var chol;
  by id;
  id month;
run;
```

C/ Repeat part A but the new data set should be in order by the variable ID

```sas
data jan_mar;
  set jan (in=_j) feb (in=_f) mar;
  by id;
  if _j then month = '1';
  else if _f then month = '2';
  else month = '3';
run;
```
#2 MATCH-MERGE DATA SETS

Use the data sets JAN, FEB, MAR from question #1. In question #1 you used TWO steps to end up with a data set that had five observations and four variables.

A/ In ONE data step, create a new data set using JAN, FEB, and MAR that has five observations (one for each ID) and four variables: ID, CHOL1, CHOL2, CHOL3.

```
proc sort data=jan;
by id;
run;

proc sort data=feb;
by id;
run;

proc sort data=mar;
by id;
run;

data jan_mar;
merge jan (rename=(chol=chol1))
feb (rename=(chol=chol2))
mar (rename=(chol=chol3))
; by id;
run;

title 'DATA SET JAN_MAR (FROM MERGE)';
proc print data=jan_mar;
var id chol1-chol3;
run;
```

B/ Same as part A, but create two data sets. One will have only those observations where an ID is present in all three months and the four variables shown in part A. The other will have observations where an ID was not present in all three months.

```
data jan_mar incomplete;
merge jan (rename=(chol=chol1) in=j)
feb (rename=(chol=chol2) in=f)
mar (rename=(chol=chol3) in=m)
; by id;
if j and f and m then output jan_mar;
else output incomplete;
run;
```